
Using the Food Patterns Equivalents (FPED) Database 
SAS Code for FPED Examples 

 
 

Overview of Examples 
 

/**************************************************************************** 
* FPED EXAMPLES - EXAMPLE 1 AND EXAMPLE 2         * 
*                     * 
* SAS program to:                * 
*                     * 
* 1) Read in the FPED total nutrient dataset.         * 
*                     * 
* 2) Estimate mean TOTAL FRUIT intake,           * 
*  stratified by MALE/FEMALE and age.          * 
*                     * 
*  EXAMPLE 1: The program will estimate the mean intakes of USDA Food  * 
*  Patterns Equivalents for TOTAL FRUIT in the WWEIA, NHANES August   * 
*  2021-August 2023 cycle among respondents 2+ years of age.  The mean  * 
*  intakes will be calculated by male/female and age and will match   * 
*  those in FPED Data Table 1a.            * 
*                     * 
* 3) Estimate mean intake of ALL FRUIT components,        * 
*  stratified by RACE/ETHNICITY and age.         * 
*                     * 
*  EXAMPLE 2: The program will estimate the mean intakes of USDA Food  * 
*  Patterns Equivalents for TOTAL FRUIT, CITRUS/MELONS/BERRIES, OTHER  * 
*  FRUIT, and FRUIT JUICE in the WWEIA, NHANES August 2021-August 2023  * 
*  cycle among respondents 2+ years of age.  The mean intakes will be  * 
*   calculated by race/ethnicity and age and will match those in     * 
*  FPED Data Table 2a.              * 
*                     * 
*  The replicated data tables (FPED Data Table 1a and 2a) are    * 
*  published at: https://www.ars.usda.gov/nea/bhnrc/fsrg/fpeddatatables. * 
*                     * 
* NOTE: There are many ways to code these FPED examples.  One approach  * 
* is being provided here, but alternative approaches are possible.   * 
****************************************************************************/ 

 

 
  



Example 1:  Replication of FPED Data Table 1A, Total Fruit, by 
Male/Female and Age 

 
/**************************************************************************** 
* FPED EXAMPLE 1 - REPLICATION OF FPED DATA TABLE 1A, TOTAL FRUIT    * 
* Stratified by Male/Female and Age           * 
****************************************************************************/ 
 
/**************************************************************************** 
* STEP 1: DOWNLOAD DATA               * 
*                     * 
* Download and extract the dataset FPED_DR1TOT_2123_sas.exe from    * 
* https://www.ars.usda.gov/northeast-area/beltsville-md-bhnrc/beltsville- * 
* human-nutrition-research-center/food-surveys-research-group/docs/fped- * 
* doc-and-datasets/. Designate a location for the file to be saved.   * 
*                     * 
* In this example, the dataset is being saved to 'C:\' but you can select * 
* a storage location convenient for you.         * 
*                     * 
* This dataset contains the day 1 total daily food pattern component      * 
* amounts for each survey respondent for foods/beverages reported in the  *  
* WWEIA, NHANES August 2021-August 2023.           * 
*                     * 
* For a complete list of variables in the dataset, see the Variable   * 
* Descriptions File at the same web location.        * 
****************************************************************************/ 
 
/**************************************************************************** 
* STEP 2: READ IN THE DATA AND RECODE VARIABLES                       * 
*                     * 
* Define a library indicating placement of extracted file.      * 
*                     * 
* Create the temporary dataset named 'data1'.            * 
*                     * 
* 1) Read in the extracted dataset 'C:\FPED_DR1TOT_2123.sas7bdat'.    * 
*                     * 
* 2) Create a variable defining the cohort of interest (incohort).  The  * 
*  cohort will include respondents 2 years of age or older     * 
*   (RIDAGEYR >= 2) with a complete, reliable dietary intake record   * 
*   (DR1DRSTZ = 1) that did not consume breastmilk (DRABF ne 1).   * 
*                     * 
* 3) Define variables capturing demographic groups of interest for    * 
*  stratifying mean intakes.  This code will produce mean estimates by * 
*  male/female and age as captured in FPED Data Table 1a.  Note that  * 
*  two different age categorizations are captured, therefore two    * 
*  age group variables will be created.         * 
****************************************************************************/ 
 
/* Designate the library for the SAS dataset.  The library should map to the  
   location where the extracted dataset was saved. */ 



libname input "C:\ "; 
 
/* Create the temporary SAS dataset work.data1. */ 
data data1; 
 
 /* Set the extracted dataset Input.FPED_DR1TOT_2123 into data1. */ 
    set Input.FPED_DR1TOT_2123; 
 
 /* Create an indicator variable named incohort which has a value of 1  
    if the respondent will be included in the analysis and 0 if not.   
    Included respondents must be 2 years of age or older (RIDAGEYR >= 2),  
    have a complete, reliable day 1 dietary intake record (DR1DRSTZ = 1), 
    and have not consumed breastmilk (DRABF ne 1). */  
 if (RIDAGEYR >= 2) and (DRABF ne 1) and (DR1DRSTZ = 1) then incohort=1; 
 else incohort=0; 
 
 /* Create the variable sex which is a copy of the original variable 
    RIAGENDR (1=male, 2=female). */ 
    sex = RIAGENDR; 
 
 /* Create the variable agegroup1 which defines the age groups of  
    interest to reproduce FPED Data Table 1a from the original  
       variable RIDAGEYR. */ 
    if       2 <= RIDAGEYR <= 5     then agegroup1 = 1; 
    else if  6 <= RIDAGEYR <= 11    then agegroup1 = 2; 
    else if 12 <= RIDAGEYR <= 19    then agegroup1 = 3; 
    else if 20 <= RIDAGEYR <= 29    then agegroup1 = 4; 
    else if 30 <= RIDAGEYR <= 39    then agegroup1 = 5; 
    else if 40 <= RIDAGEYR <= 49    then agegroup1 = 6; 
    else if 50 <= RIDAGEYR <= 59    then agegroup1 = 7; 
    else if 60 <= RIDAGEYR <= 69    then agegroup1 = 8; 
    else if RIDAGEYR >= 70          then agegroup1 = 9; 
 
  
 /* Create the variable agegroup2 which defines additional age groups  
    of interest (all children and all adults) to reproduce FPED Data  
    Table 1a from the original variable RIDAGEYR. */ 
    if      ( 2 <= RIDAGEYR <= 19)   then agegroup2 = 1; 
    else if (RIDAGEYR >= 20)         then agegroup2 = 2; 
run; 
 
/**************************************************************************** 
* STEP 3: GENERATE THE WEIGHTED MEAN INTAKES OF USDA FOOD PATTERNS    * 
* COMPONENTS                 * 
*                     * 
* IMPORTANT CODE ELEMENTS OF THE SURVEYMEANS PROCEDURE:      * 
*                     * 
* 1) PROC SURVEYMEANS: Calculates means and standard errors while allowing  * 
*    for a complex survey design.  The dietary component of NHANES,   * 
*  WWEIA is generated from a complex, multistage, probability sampling * 
*    design.                * 



*                     * 
* 2) STRATA: Indicates the strata variable for a complex sample design  * 
*  (named SDMVSTRA in the downloaded data).        * 
*                     * 
* 3) CLUSTER: Indicates the cluster variable for a complex sample design * 
*  (named SDMVPSU).              * 
*                     * 
* 4) WEIGHT: Indicates the sampling weight variable (use WTDRD1, the day 1 * 
*    dietary sample weight.             * 
*                     * 
* 5) DOMAIN: Indicates the subpopulations desired for analysis.  A   * 
*  single variable can be listed here. For example, by listing incohort * 
*  below, SAS will produce estimates for each level of incohort (i.e., * 
*  where incohort=1 and where incohort=0). Cross-products can also be  * 
*  listed.  For example, by listing incohort*sex, SAS will produce  * 
*   estimates separately for males and females both where incohort=1 and * 
*  where incohort=0.              * 
*                     * 
* 6) VAR: Lists the variables for which means (and other statistics) will * 
*  be calculated. The variable for identifying the total fruit component * 
*    in FPED Data Table 1a is used (DR1T_F_TOTAL).       * 
*                     * 
* 7) ODS OUTPUT DOMAIN=dataset_name: Names an output dataset that will   * 
*  store the estimates for the variables listed in the VAR statement   * 
*  within the subpopulations specified in the DOMAIN statement.   * 
*                     * 
*   For each output dataset, a modified copy will be created that    * 
*  contains only those results for incohort=1.  In addition, it will   * 
*  add two common variables, agegroup and nsex, that capture the   * 
*  desired age and sex groups as displayed in Table 1a.  In other words: * 
*                     * 
  
*  agegroup = 1 for 2-5 year olds   nsex = 1 for males    * 
*   agegroup = 2 for 6-11 year olds  nsex = 2 for females   * 
*  agegroup = 3 for 12-19 year olds  nsex = 3 for males and females * 
*  agegroup = 4 for 20-29 year olds          * 
*  agegroup = 5 for 30-39 year olds          * 
*  agegroup = 6 for 40-49 year olds          * 
*  agegroup = 7 for 50-59 year olds           * 
*  agegroup = 8 for 60-69 year olds          * 
*  agegroup = 9 for 70+ year olds           * 
*  agegroup = 10 for 2-19 year olds          * 
*  agegroup = 11 for 20+ year olds          *  
*  agegroup = 12 for 2+ year olds           * 
****************************************************************************/ 
 
/* Sort the data by strata (SDMVSTRA) and PSU (SDMVPSU). */ 
proc sort data=data1; 
    by SDMVSTRA SDMVPSU; 
run; 
 



/* All included respondents. */ 
proc surveymeans nobs mean stderr data = data1; 
    strata SDMVSTRA;    
    cluster SDMVPSU;   
    weight WTDRD1; 
    domain incohort; 
    var DR1T_F_TOTAL; 
 ods output domain=est_all; 
run; 
/* Create a dataset estimates_all which retains only the estimates for the 
   cohort of interest (incohort=1) from the dataset est_all.  Add the 
   variables agegroup (set value to 12 since age is not in the domain 
   statement and thus all respondents 2+ are included) and nsex (set  
   value to 3 since sex is not in the domain statement and thus  
   estimates reflect males and females combined). */ 
data estimates_all; 
   set est_all; 
   if incohort=1; 
   agegroup = 12; 
   nsex=3; 
run; 
 
/* All included respondents by agegroup2. */ 
proc surveymeans nobs mean stderr data = data1; 
    strata SDMVSTRA;    
    cluster SDMVPSU;   
    weight WTDRD1; 
    domain incohort*agegroup2; 
    var DR1T_F_TOTAL; 
 ods output domain=est_age2; 
run; 
/* Create a dataset estimates_age2 which retains only the estimates for the 
   cohort of interest (incohort=1) from the dataset est_age2.  Add the 
   variables agegroup (set values to 10 and 11 since the variable agegroup2 
   is in the domain statement, indicating age groups 2-19 and 20+) and  
   nsex (set value to 3 since sex is not in the domain statement and thus  
   estimates reflect males and females combined). */ 
data estimates_age2; 
   set est_age2; 
   if incohort=1; 
   if agegroup2=1 then agegroup = 10; 
 if agegroup2=2 then agegroup = 11; 
   nsex=3; 
run; 
 
/* All included respondents by sex. */ 
proc surveymeans nobs mean stderr data = data1; 
    strata SDMVSTRA;    
    cluster SDMVPSU;   
    weight WTDRD1; 
    domain incohort*sex; 



    var DR1T_F_TOTAL; 
 ods output domain=est_sex; 
run; 
/* Create a dataset estimates_sex which retains only the estimates for the 
   cohort of interest (incohort=1) from the dataset est_sex.  Add the 
   variables agegroup (set value to 12 since age is not in the domain 
   statement and thus all respondents 2+ are included) and nsex (set  
   value to match sex since sex is included in the domain statement,  
   indicating sex groups males and females). */ 
data estimates_sex; 
   set est_sex; 
   if incohort=1; 
   agegroup = 12; 
   nsex=sex; 
run; 
 
/* All included respondents by sex and agegroup1. */ 
proc surveymeans nobs mean stderr data = data1; 
    strata SDMVSTRA;    
    cluster SDMVPSU;   
    weight WTDRD1; 
    domain incohort*sex*agegroup1; 
    var DR1T_F_TOTAL; 
 ods output domain=est_sexage1; 
run; 
 
/* Create a dataset estimates_sexage1 which retains only the estimates for the 
   cohort of interest (incohort=1) from the dataset est_sexage1.  Add the 
   variables agegroup (set values to match agegroup1 since agegroup1 is  
   included in the domain statement, indicating age groups 2-5, 6-11,..., 
   70+) and nsex (set value to match sex since sex is included in the  
   domain statement, indicating sex groups males and females). */ 
data estimates_sexage1; 
   set est_sexage1; 
   if incohort=1; 
   agegroup = agegroup1; 
   nsex=sex; 
run; 
 
/* All included respondents by sex and agegroup2. */ 
proc surveymeans nobs mean stderr data = data1; 
    strata SDMVSTRA;    
    cluster SDMVPSU;   
    weight WTDRD1; 
    domain incohort*sex*agegroup2; 
    var DR1T_F_TOTAL; 
 ods output domain=est_sexage2; 
run; 
/* Create a dataset estimates_sexage2 which retains only the estimates for the 
   cohort of interest (incohort=1) from the dataset est_sexage2.  Add the 
   variables agegroup (set values to 10 and 11 since the variable agegroup2 



   is in the domain statement, indicating age groups 2-19 and 20+) and nsex  
   (set value to match sex since sex is included in the domain statement,    
   indicating sex groups males and females). */ 
data estimates_sexage2; 
  set est_sexage2; 
   if incohort=1; 
   if agegroup2=1 then agegroup = 10; 
 if agegroup2=2 then agegroup = 11; 
   nsex=sex; 
run; 
 
/**************************************************************************** 
* STEP 4: CREATE FORMATS FOR THE DEMOGRAPHIC GROUPS OF INTEREST             * 
*                     * 
* Create formats for the new variables agegroup and nsex.  They are   * 
* called agegroupf and sexf and match the categories from Table 1a.   * 
****************************************************************************/ 
 
proc format; 
    value agegroupf 
        1   = " 2 -  5    " 
        2   = " 6 - 11" 
        3   = "12 - 19" 
        4   = "20 - 29" 
        5   = "30 - 39" 
        6   = "40 - 49" 
        7   = "50 - 59" 
        8   = "60 - 69" 
        9   = "70 and over" 
        10  = "2-19" 
        11  = "20 and over" 
  12  = "2 and over"; 
 
    value sexf  
        1   = "Males  " 
        2   = "Females" 
  3   = "Males and females"; 
run; 
 
/**************************************************************************** 
* STEP 5: CREATE PERMANENT SAS DATASET OF RESULTS MATCHING TABLE 1A   * 
*                     * 
* Prepare a permanent SAS dataset that contains the information from  * 
* and matches the structure of FPED Data Table 1a.        * 
****************************************************************************/ 
 
/* Stack the output datasets from Step 3 on top of one another and store in  
   a new dataset, estimates_stacked. The dataset estimates_stacked has one  
   row per subpopulation (identified by the variables sex and agegroup) and  
   contains the sample size, mean, and standard error of the of USDA Food  
   Patterns Equivalents for total fruit, as captured in FPED Data Table 1a. */ 



data estimates_stacked; 
   set estimates_all estimates_age2 estimates_sex estimates_sexage1  
        estimates_sexage2; 
   drop domainlabel incohort agegroup1 agegroup2 sex; 
run; 
 
/* Sort the data to match the order of FPED Data Table 1a. */ 
proc sort data=estimates_stacked; 
    by nsex agegroup; 
run; 
 
/* Create a final permanent SAS dataset 'input.FPED_Ex1' with the same  
   structure as FPED Data Table 1a that is stored in the same location as  
   the original downloaded data. */ 
data input.FPED_Ex1; 
   retain nsex agegroup N mean stderr; 
   set estimates_stacked; 
 
 *Remove variables no longer needed; 
   drop VarName VarLabel; 
 
 *Apply formats to nsex and agegroup variables; 
    format nsex sexf. agegroup agegroupf.; 
 
 *Present mean/stderr values to two/three decimal places; 
 format mean 8.2 stderr 8.3; 
run; 
 
/**************************************************************************** 
* STEP 6: EXPORT A COPY OF THE FINAL DATA INTO EXCEL (OPTIONAL)    * 
****************************************************************************/ 
 
/* Export a copy of the final SAS dataset into an excel spreadsheet. The use 
of ods excel allows for retention of the SAS formats. */ 
ods excel file="C:\FPED_Ex1.xlsx"; 
 ods excel options(sheet_name="Table 1A, Total Fruit"); 
 proc print data=input.FPED_Ex1 noobs; 
  title "Mean Cup Equivalents of Total Fruit Consumed Per Individual,  
     by Male/Female and Age, in the United States,  
  August 2021-August 2023"; 
 run; 

    ods excel close; 

  



Example 2:  Replication of FPED Data Table 2A, All Fruit Components, by 
Race/Ethnicity and Age 

 
/**************************************************************************** 
* FPED EXAMPLE 2 - REPLICATION OF FPED DATA TABLE 2A, TOTAL FRUIT,    * 
* CITRUS/MELONS/BERRIES, OTHER FRUIT, and FRUIT JUICE      * 
* Stratified by Race/Ethnicity and Age          * 
****************************************************************************/ 
 
/**************************************************************************** 
* STEP 1: DOWNLOAD DATA -- SEE EXAMPLE 1          * 
****************************************************************************/ 
 
/**************************************************************************** 
* STEP 2: READ IN THE DATA AND RECODE VARIABLES -- SEE EXAMPLE 1,        * 
* MODIFYING PART 3 AND ADDING PART 4 AS FOLLOWS         * 
*                     * 
* 3) Define variables capturing demographic groups of interest for    * 
*  stratifying mean intakes.  This code will produce mean estimates by * 
*  race/ethnicity and age as captured in FPED Data Table 2a.    * 
*                     * 
* 4) Label the variables capturing the FPED components of interest   * 
*  (DR1T_F_TOTAL, DR1T_F_CITMLB, DR1T_F_OTHER, DR1T_F_JUICE).    *  
****************************************************************************/ 
 
/* Designate the library for the SAS dataset.  The library should map to the  
   location where the extracted dataset was saved. */ 
libname input "C:\ "; 
 
/* Create the temporary SAS dataset work.data2. */ 
data data2; 
 
 /* Set the extracted dataset Input.FPED_DR1TOT_2123 into data2. */ 
    set Input.FPED_DR1TOT_2123; 
 
 /* Create an indicator variable named incohort which has a value of 1  
    if the respondent will be included in the analysis and 0 if not. */  
 if (RIDAGEYR >= 2) and (DRABF ne 1) and (DR1DRSTZ = 1) then incohort=1; 
 else incohort=0; 
 
 /* Create a variable capturing the desired age group categories. */ 
    if      ( 2 <= RIDAGEYR <= 19)   then agegroup2 = 1; 
    else if (RIDAGEYR >= 20)         then agegroup2 = 2; 
 
/* Create the variable race_eth which defines the race/ethnicity groups of    
   interest to reproduce FPED Data Table 2a from the original variable  
   RIDRETH3. */ 
    if      RIDRETH3 = 3            then race_eth = 1;   * Non-Hisp White; 
    else if RIDRETH3 = 4            then race_eth = 2;   * Non-Hisp Black; 
    else if RIDRETH3 = 6            then race_eth = 3;   * Non-Hisp Asian; 



    else if RIDRETH3 in(1 2)        then race_eth = 4;   * Hispanic      ; 
    else                                 race_eth = 5;   * Other Race    ; 
 
 /* Create labels for the variables capturing the desired FPED 
    components, all fruits.  These will be used to generate variable 
    names later in the code. */ 
    label DR1T_F_TOTAL          = "Total_Fruit"; 
    label DR1T_F_CITMLB         = "Citrus_Melons_Berries"; 
    label DR1T_F_OTHER          = "Other_Fruit"; 
    label DR1T_F_JUICE          = "Fruit_Juice"; 
run; 
 
/**************************************************************************** 
* STEP 3: GENERATE THE WEIGHTED MEAN INTAKES OF USDA FOOD PATTERNS    * 
* COMPONENTS -- SEE EXAMPLE 1, MODIFYING PART 6 AND THE FINAL PARAGRAPH  * 
* AS FOLLOWS:                * 
*                     * 
* 6) VAR: Lists the variables for which means (and other statistics) will * 
*  be calculated. Because we are reproducing FPED Data Table 2a for all * 
*  fruit, the four variables identifying fruit components are listed   * 
*   (DR1T_F_TOTAL, DR1T_F_CITMLB, DR1T_F_OTHER, DR1T_F_JUICE).    * 
*                     * 
*   For each output dataset, a modified copy will be created that    * 
*  contains only those results for incohort=1.  It will also exclude   * 
*  race/ethnicity category 'other' (race_eth=5) due to the small sample * 
*   size and add a common variable, agegroup, that capture the age groups * 
*  displayed in Table 2a.  In other words:        * 
*                     * 
*  agegroup = 1 for 2-19 year olds   race_eth = 1 for Non-Hispanic White  * 
*   agegroup = 2 for 20+ year olds    race_eth = 2 for Non-Hispanic Black  * 
*  agegroup = 3 for 2+ year olds    race_eth = 3 for Non-Hispanic Asian * 
*             race_eth = 4 for Hispanic   * 
****************************************************************************/ 
 
/* Sort the data by strata and PSU. */ 
proc sort data=data2; 
    by SDMVSTRA SDMVPSU; 
run; 
 
/* All included respondents by race/ethnicity. */ 
proc surveymeans nobs mean stderr data = data2; 
    strata SDMVSTRA;    
    cluster SDMVPSU;   
    weight WTDRD1; 
    domain incohort*race_eth; 
    var DR1T_F_TOTAL DR1T_F_CITMLB DR1T_F_OTHER DR1T_F_JUICE; 
 ods output domain=est_race; 
run; 
/* Create a dataset estimates_race which retains only the estimates for the 
   cohort of interest (incohort=1) from the dataset est_race and excludes 
   the estimates for race=5.  Add the variable agegroup (set value to 3  



   since age is not in the domain statement and thus all respondents 2+ are     
   included). */ 
data estimates_race; 
   set est_race; 
   if incohort=1 and race_eth^=5; 
   agegroup=3; 
run; 
 
/* All included respondents by race/ethnicity and agegroup2. */ 
proc surveymeans nobs mean stderr data = data2; 
    strata SDMVSTRA;    
    cluster SDMVPSU;   
    weight WTDRD1; 
    domain incohort*race_eth*agegroup2; 
    var DR1T_F_TOTAL DR1T_F_CITMLB DR1T_F_OTHER DR1T_F_JUICE; 
 ods output domain=est_raceage2; 
run; 
/* Create a dataset estimates_raceage2 which retains only the estimates for the 
   cohort of interest (incohort=1) from the dataset est_raceage2 and  
   excludes the estimates for race_eth=5.  Add the variable agegroup (retain  
   the values from agegroup2 since agegroup2 is in the domain statement,  
   indicating age groups 2-19 and 20+). */ 
data estimates_raceage2; 
   set est_raceage2; 
   if incohort=1 and race_eth^=5; 
   agegroup = agegroup2; 
run; 
 
/**************************************************************************** 
* STEP 4: CREATE FORMATS FOR THE DEMOGRAPHIC GROUPS OF INTEREST    * 
*                     * 
* Create formats for the new variables agegroup and race.  They are   * 
* called agegroupf and racef and match the categories from Table 2a.  * 
****************************************************************************/ 
 
proc format; 
    value agegroupf 
        1  = "2-19       " 
        2  = "20 and over" 
  3  = "2 and over"; 
 
    value race_ethf  
        1   = "Non-Hispanic White" 
        2   = "Non-Hispanic Black" 
        3   = "Non-Hispanic Asian" 
        4   = "Hispanic"; 
run; 
 
/**************************************************************************** 
* STEP 5: CREATE PERMANENT SAS DATASET OF RESULTS MATCHING TABLE 2A   * 
*                     * 



* Prepare a permanent SAS dataset that contains the information from  * 
* and matches the structure of FPED Data Table 2a.        * 
****************************************************************************/ 
 
/* Stack the output datasets from Step 3 on top of one another and store in a new 
   dataset estimates_stacked. */ 
data estimates_stacked; 
   set estimates_race estimates_raceage2; 
   drop domainlabel incohort agegroup2; 
run; 
 
/* Sort the data to match the order of FPED Data Table 2a. */ 
proc sort data=estimates_stacked; 
    by race_eth agegroup VarLabel; 
run; 
 
/* Transpose the variable 'N' from 'estimates_stacked' which contains the  
   sample size.  Name the transposed values using the prefix  
   designation before the variable in the ID line (i.e., N_VarLabel). 
   Race/ethnicity and age groups are retained, yielding one row per  
   race/ethnicity and agegroup combination and columns for all  
   fruit component sample sizes. */  
proc transpose data=estimates_stacked out=FPED_N prefix=N_; 
   by race_eth agegroup; 
   var N; 
   id VarLabel; 
run; 
 
/* Repeat for the means. */ 
proc transpose data=estimates_stacked out=FPED_mean prefix=Mean_; 
   by race_eth agegroup; 
   var mean; 
   id VarLabel; 
run; 
 
/* Repeat for the standard errors. */ 
proc transpose data=estimates_stacked out=FPED_stderr prefix=stderr_; 
   by race_eth agegroup; 
   var stderr; 
   id VarLabel; 
run; 
 
/* Merge the three datasets containing the sample size, means, and 
   standard errors together by race/ethnicity and agegroup, dropping  
   unneeded variables. */ 
data FPED; 
   merge FPED_N FPED_mean FPED_stderr; 
   by race_eth agegroup; 
   drop _NAME_ _LABEL_; 
run; 
 



/* Create a final permanent SAS dataset 'input.FPED_Ex2' with the same  
   structure as FPED Data Table 2a that is stored in the same location as  
   the original downloaded data. */ 
data input.FPED_Ex2; 
   retain race_eth agegroup N_total_fruit mean_total_fruit stderr_total_fruit 
   mean_citrus_melons_berries stderr_citrus_melons_berries 
   mean_other_fruit stderr_other_fruit 
   mean_fruit_juice stderr_fruit_juice; 
   set FPED; 
 
 *Apply a common sample size name (N) to the sample size variable; 
   rename N_total_fruit=N; 
 
 *Remove variables no longer needed; 
  drop N_citrus_melons_berries N_fruit_juice N_Other_fruit; 
 
 *Apply formats to nsex and agegroup variables; 
    format race_eth race_ethf. agegroup agegroupf.; 
 
 *Present mean values to two decimal places and standard errors to  
 three decimal places; 
 format mean_total_fruit 8.2 mean_citrus_melons_berries 8.2 
      mean_other_fruit 8.2 mean_fruit_juice 8.2  
      stderr_total_fruit 8.3 stderr_citrus_melons_berries 8.3 
      stderr_other_fruit 8.3 stderr_fruit_juice 8.3; 
run; 
 
/**************************************************************************** 
* STEP 6: EXPORT A COPY OF THE FINAL DATA INTO EXCEL (OPTIONAL)    * 
****************************************************************************/ 
 
/* Export a copy of the final SAS dataset into an excel spreadsheet. The use 
of ods excel allows for retention of the SAS formats. */ 
ods excel file="C:\FPED_Ex2.xlsx"; 
 ods excel options(sheet_name="Table 2A, Total Fruit"); 
 proc print data=input.FPED_Ex2 noobs; 
  title "Mean Cup Equivalents of Total Fruit, Citrus/Melons/Berries,  
  Other Fruit, and Fruit Juice Consumed per Individual, 
  by Race/Ethnicity and Age, in the United States,  
  August 2021-August 2023"; 
 run; 
ods excel close; 

